
Hitachi HM55B Compass Module (#29123) 
General Description 
The Hitachi HM55B Compass Module is a dual-axis magnetic field sensor that can add a sense of direction 
to your next electronic or robotic project.  The sensing device on the Compass Module is an Hitachi 
HM55B chip.  An onboard regulator and resistor protection make the 3 V HM55B chip compatible with 5 V 
BASIC Stamp supply and signal levels.  The Compass Module also makes all the power and signal 
connections on the tiny surface mount HM55B chip accessible in a breadboard-friendly 0.3 inch wide 6-
pin DIP package.  Acquiring measurements from the module is made easy with the BASIC Stamp 2 
microcontroller's SHIFTIN and SHIFTOUT commands, which are designed for synchronous serial 
communication with chips like the HM55B.   

Features 
• Sensitive to microtesla (µT) variations in magnetic field strength 
• Simplifies direction by resolving magnetic field measurements into two components axes  
• 8-bit angular resolution +/- 1 brad after calibration 
• Only 30 to 40 ms between start measurement and data-ready 
• Built-in resistor protection for data pins eliminates bus conflict risks 
• Compact and breadboard-friendly 0.3 inch, 6-pin DIP package 
• Compatible with all BASIC Stamp, Javelin Stamp and SX microcontrollers 
• Especially easy to control, read, and determine direction with the BASIC Stamp 2 series 

microcontroller and the PBASIC SHIFTOUT, SHIFTIN, and ATN commands 

Application Ideas 
• Mobile robot direction sensor 
• Handheld electronic compass 
• Weathervane indicator for remote weather stations 
• Audible compass for the vision impaired 
• Automotive electronic compass 

Quick Start Circuit 

   29123 

Connecting and Testing 
√ Build the circuit shown on page 1. 
√ Place your board on a level surface away from magnetic field disturbances such as motors, 

magnets, and large metal objects.  Also, make sure your programming cable and power cords do 
not wrap around or pass near the sensor. 

√ Enter and run the BASIC Stamp 2 test program on pages 3-4.  This program is also available 
from the Downloads section of the Hitachi HM55B Compass Module product page: 

 
       http://www.parallax.com/detail.asp?product_id=29123  
 

√ Test your Compass Module for direction detection as shown here.  If you notice a few degrees 
variation from true magnetic north, try the calibration section below.   

 

 

Calibration  
√ With the help of a magnetic compass, align your Compass Module (or the edge of the board it's 

on) to magnetic north. 
√ Make a note of the y-axis reading.  If it's not zero, enter the value reported by the Debug 

Terminal into the test program's YOffset CON directive. 
√ Rotate your board so that it is pointing west, and repeat for the process for the x-axis and 

XOffset constant. 
√ Run the modified program and test the various direction measurements.  The direction 

measurements should now be significantly improved. 
 
You can examine the measurement resolution more closely by commenting the line in the program that 
converts from binary radians to degrees.  Also, modify the DEBUG command so that it displays the label 
"brads" instead of "degrees".  Running this further modified program will result in the binary radian 
representation of the direction.  0 will still be north, northwest will be 32, west will be 64, southwest 96, 
south 128, and so on. 

Sensitivity  
The magnetic field strength corresponding to a measurement of 1 should resolve to somewhere between 
1 and 1.6 µT.  To find out how many microteslas per unit each axis reports with your particular Compass 
Module, start by finding the maximum possible axis measurement.  Do this by orienting the axis to 
magnetic north, then tilt up and down until you find the highest value.  Compare this to the total 
magnetic field intensity in your area.  In the continental US, you can find a Total Intensity map on this 
page http://nationalatlas.gov/articles/geology/a_geomag.html.  Make sure to divide their nanotesla (nT) 
values by 1000 to convert to microteslas (µT). 

BASIC Stamp® 2, 2sx, 2p, 2pe, and 2px Microcontroller Test Program 
If you are using a BS2sx, p, e, pe, or px, make sure to update the $STAMP directive before running the 
program.  No other part of the program will need to be changed.  
 
' ============================================================================ 
' TestDigitalCompass.bs2 - This Hitachi HM55B Compass Module test program 
' displays x (N/S) and y (W/E) axis measurements and degrees from north. 
' 
' Author.... (C) 2005 Parallax, Inc -- All Rights Reserved 
' Email.....  support@parallax.com 
' 
' {$STAMP BS2} 
' {$PBASIC 2.5}   
' ============================================================================ 
 
' -----[ Pins/Constants/Variables ]------------------------------------------- 
DinDout        PIN      6                    ' P6 tranceives to/from Din/Dout 
Clk            PIN      5                    ' P5 sends pulses to HM55B's Clk 
En             PIN      4                    ' P4 controls HM55B's /EN(ABLE) 
 
YOffset        CON      0                    ' Enter measured y at north here 
XOffset        CON      0                    ' Enter measured x at west here 
 
Reset          CON      %0000                ' Reset command for HM55B 
Measure        CON      %1000                ' Start measurement command 
Report         CON      %1100                ' Get status/axis values command 
Ready          CON      %1100                ' 11 -> Done, 00 -> no errors 
NegMask        CON      %1111100000000000    ' For 11-bit negative to 16-bits 
 
x              VAR      Word                 ' y-axis data 
y              VAR      Word                 ' x-axis data 
status         VAR      Nib                  ' Status flags 
angle          VAR      Word                 ' Store angle measurement 
 
' -----[ Main Routine ]------------------------------------------------------- 
 
DO                                           ' Main loop 
 
  GOSUB Get_Compass_Axes                     ' Get x, and y values 
 
  angle = x ATN y                            ' Convert x and y to brads 
  angle = angle */ 360                       ' Convert brads to degrees 
 
  DEBUG HOME,                                ' Display axes and degrees 
        "x-axis N(-S) = ",SDEC x, CLREOL, 
        CR, "y-axis W(-E) = ",SDEC y,CLREOL, 
        CR, CR, DEC3 angle, " Degrees" 
 
  PAUSE 200                                  ' Debug delay for slower PCs 
 
LOOP                                         ' Repeat main loop 

' -----[ Subroutines ]-------------------------------------------------------- 
 
Get_Compass_Axes:                            ' Compass subroutine 
 
  HIGH En: LOW En                            ' Disable/enable module 
  SHIFTOUT DinDout,clk,MSBFIRST,[Reset\4]    ' Send reset command 
  HIGH En: LOW En                            ' Disable/enable module 
  SHIFTOUT DinDout,clk,MSBFIRST,[Measure\4]  ' Start measurement 
  status = 0                                 ' Clear previous status flags 
  DO                                         ' Status flag checking loop 
    HIGH En: LOW En                          ' Disable/enable sensor 
    SHIFTOUT DinDout,clk,MSBFIRST,[Report\4] ' Request status 
    SHIFTIN  DinDout,clk,MSBPOST,[Status\4]  ' Get Status 
  LOOP UNTIL status = Ready                  ' Exit loop when status is ready 
  SHIFTIN  DinDout,clk,MSBPOST,[x\11,y\11]   ' Get x & y axis values 
  HIGH En                                    ' Disable module 
 
  IF (y.BIT10 = 1) THEN y = y | NegMask      ' Store 11-bits as signed word 
  IF (x.BIT10 = 1) THEN x = x | NegMask      ' Repeat for other axis 
 
  y = y - YOffset                            ' Correct y-axis offset 
  x = x - XOffset                            ' Correct x-axis offset 
 
  RETURN 

Theory of Operation 
The Hitachi HM55B Compass Module has two axes, x and y.  Each axis reports the strength of the 
magnetic field's component parallel to it.  The x-axis reports (field strength) × cos(θ), and the y-axis 
reports the (field strength) × sin(θ).  To resolve θ into an angle from north, use arctan(y/x), which in 
PBASIC 2.5 is x ATN y.  The ATN command returns the angle in binary radians, which parse a full circle 
into 256 equal segments.  To convert to degrees (360 equal segments) with PBASIC, just apply */ 360 to 
the variable storing the binary radian measurement. 
 

 
 
The Hitachi HM55B chip on the Hitachi HM55B Compass Module reports its x and y axis measurements in 
terms or microteslas (µT) in 11-bit signed values.  The HM55B is designed to return a value of 1 for a 
north magnetic field of 1 µT parallel to one of its axes.  If the magnetic field is south (north in the 
opposite direction), the value will be -1.  Keep in mind that these are nominal values.  According to the 
HM55B datasheet, the actual µT value for a measurement of 1 could range anywhere from 1 to 1.6 µT.  
Also keep in mind that a negative 11-bit value will not appear negative in a word variable unless a mask 
is applied.  For example, when bit-10 is 1, bits 11 to 15 are also changed to 1 with a mask in the test 
program.   
 
The microcontroller connected to the HM55B must control its enable and clock inputs and use 
synchronous serial communication to get the axis measurements from its data input and data output 
pins.  For example, a BASIC Stamp 2 can be programmed to control the Compass Module's enable lines 
with HIGH/LOW and send values that reset the device and start a measurement with SHIFTOUT 
commands.  The SHIFTOUT command controls the clock input as it sends data bit values to the Compass 
Module's data input.  The converse of SHIFTOUT is SHIFTIN, which also controls the device's clock input 
as it collects data bits sent by the device's data output pin.   
 
It takes the HM55B 30 to 40 ms to complete a given measurement.  The microcontroller can either 
perform other tasks during this time or poll until the measurement is complete.  The polling is a 
combination of SHIFTOUT commands that request the status.  When the SHIFTIN receives status flags 
indicating that the measurement is complete, a second SHIFTIN command can then store the 11-bit x 
and y axis measurements in variables. 

Precautions 
• Do not operate or store the Compass Module near sources of strong magnetic fields.  Strong 

magnetic fields can be created by bar and ring magnets, electric motors, and other coil elements 
such as solenoids, relays, and large inductors.  

• Do not apply magnetic fields in excess of 300 µT to the Compass Module.  Magnetic fields 
stronger than 300 µT can permanently damage the sensor. 

• Mount the Compass Module as far away as possible from magnetic field disturbances.  These 
include magnets, motors, power cords, coils, metal boxes, and sometimes the ground.  

• Do not apply voltages to the device that are outside the values stated in the Pin Definitions and 
Ratings section. 

Angle θ = arctan(y/x)

 Specifications 

Symbol Quantity Minimum Typical Maximum Units 

BSE Sensitivity†  1.0 1.6 uT/lsb 

H Linear measurement range† -180  180 uT 

dθ Axis offset   20 ° 

TCONV Conversion time†  30 40 ms 

TOPE Operating temperature 0  70 °C 
        † From Hitachi HM55B Datasheet 

Pin Definitions and Ratings 

 
 

 

 

Symbol Quantity Minimum Typical Maximum Units 

Vcc Supply Voltage 4.8 5.0 5.2 V 

Icc(Ave) Average active supply current*  5 7 mA 

Icc(Pk) Peak instantaneous current**  30 45 mA 

Icc(Sb) Standby supply current  2 3 mA 

GND Ground reference connection  0  V 

VOH Signal high transmit (Dout) Vcc × 0.9 Vcc Vcc + 0.5 V 

VOL Signal low transmit (Dout) GND - 0.3 GND Vcc × 0.15 V 

VIH Signal high receive (/En, CLK, Din) Vcc × 0.8 Vcc Vcc + 0.3 V 

VIL Signal low receive (/En, CLK, Din) - 0.3 GND Vcc × 0.12 V 

        *   Measurement cycle = 80 ms 
        ** Typical duration is 5 µs 

(1) Din  -  Serial data input 
(2) Dout  -  Serial data output 
(3) GND  -  Ground -> 0 V 
(4) CLK  -  Synchronous clock input 
(5) /EN  -  Active-low device enable 
(6) Vcc  -  +5 V power input 

29123 

Connection Diagrams 
The 3-wire interface is recommended for most applications.  While all the connections shown here are to 
individual I/O pins, the /EN pin is the only one that needs a dedicated I/O pin.  The Din/Dout pins can 
share a line with other synchronous serial devices, and likewise with CLK.   
 

 
 
The Din and Dout pins do not have to be tied together; they can also be controlled individually.  This 
makes it possible to share communication lines with other synchronous serial devices that have dedicated 
input and output lines. 
 

 

Command Set 
These commands are shifted-out to the Compass Module.   
 

Binary Value Quantity 

0000 Reset device 

0001 Start measurement 

0011 Report measurement status (and transmit the measurement if it's ready) 

 

4-wire interface

3-wire interface

Status Flags 
The Compass Module will reply to the report measurement status command with one of these values. 
 

Binary Value Quantity 

Bits - 3210  3 and 2 indicate measurement completion, 1 and 0 indicate measurement errors 

1100 11 -> Measurement completed; 00 -> no errors 

00XX Measurement still in progress, or the device has been reset. 

XX11 /EN did not receive low-high-low signal between start and report commands 

Communication Protocol 
All values transmitted to and received from the Compass Module are most significant bit first, with the bit 
value valid after the clock signal's rising edge.  For the BASIC Stamp 2, this means set the Mode 
argument to MSBFIRST for SHIFTOUT and MSBPOST for SHIFTIN. 
 
To reset the HM55B, take /EN from high to low, and shift-out %0000, then set /EN high again. 
 

 
 
After reset, start a measurement by taking /EN low again, then shift out %1000.  Leave /EN low until 
checking the measurement status. 

To check the measurement status, start by sending a positive pulse to /EN.  Then, shift-out %1100, and 
shift-in the status flags.  While the measurement is in progress, the end flag and error flag will both be 
00.  The device may be polled for status repeatedly until the measurement is complete, at which point 
the end flag will change to 11.  Upon receipt of %1100, discontinue polling.  Leave /EN low, and move on 
to shifting-in the x and y-axis values. 
 

 
 
Shifting-in the x and y-axis values is a simple matter of shifting-in 11 bits for the x-axis measurement 
followed by 11 more bits for the y-axis measurement.   

Module Dimensions 

 

Module Schematic 

 

Resources and Downloads 
Check out the Parallax Digital Compass Sensor product page for example programs, articles, Stamps in 
Class activities, the HM55B datasheet, and more: 

 
       http://www.parallax.com/detail.asp?product_id=29123  

 


